

4.3 Windung schaltungsgespeist

Eine Windung (massiver Leiter) wird von einem aufgeladenen Kondensator über einen Widerstand gespeist. Das Problem wird rotationssymmetrisch zweidimensional behandelt. Wegen der Symmetrie wird nur die obere Hälfte betrachtet. Abb. 4.3-1 zeigt die Geometrie und die Berechnungsebene.




Abb. 4.3-1 Geometrie

Die verwendeten Abmessungen sind den ersten Programmzeilen entnehmbar. Zuerst wird die Geometrie generiert und vernetzt, ohne die speisende Schaltung zu berücksichtigen.

Programm

```
!eine Windung (Kupfer) stromdurchflossen
!zweidimensional axialsymmetrisch
!Ausnutzung von Symmetrie
!schaltungsgespeist (Kondensator)
wiradin=35e-3           !Windungs-Radius innen
wiradau=40e-3           !Windungs-Radius außen
wiho=40e-3              !Windungs-Höhe
beradin=60e-3           !Bereichs-Radius innen
beradau=150e-3          !Bereichs-Radius außen
gacu=57e6                !Gamma Kupfer
mw=.25e-3               !Maschenweite
mw=.5e-3
/prep7
et,1,plane53,,1         !Luft
et,2,plane53,4,,1       !Leiter massiv
et,3,infin110,,1        !Randelement
mp,murx,1,1             !Luft
mp,murx,2,1             !Kupfer
mp,rsvx,2,1/gacu
```

```

r,1,(wiradau-wiradin)*wiho/2 !Wicklungsfläche
!----Geometrie-----
pcirc,0,beradau,0,90
pcirc,0,beradin,0,90
rectng,wiradin,wiradau,0,wiho/2
aovlap,all
!A3:Leiter A5:Bereich innen A4:Bereich außen
!-----Vernetzung--Leiter-----
mshape,0,2d !Vernetzen mit Vierecken
mshkey,1
esize,mw !Maschenweite
type,2 $mat,2 $real,1
amesh,3
!----Vernetzung---Luft-----
csys,1
lsel,s,loc,x,beradin,beradau
lsel,u,loc,x,beradin
lsel,u,loc,x,beradau
lesize,all,,,4,1/4
esize,,8
type,1 $mat,1
amesh,4 !Luft außen
mshkey,0
esize,wiho/4
type,1 $mat,1
amesh,5 !Luft innen
!----Rand-----
nsel,s,loc,x,beradau
esln
emodif,all,type,3 !Randelemente
sf,all,inf
allsel

```

Nun wird die speisende Schaltung generiert. Dazu werden zuerst die Bauelemente-parameter vereinbart.

```

!---speisende Schaltung-----
kapazi=10e-6 !Kapazität
u0=-10e3 !Ladespannung
rvor=10e-3 !Vorwiderstand

```

Die für die diskreten Bauelemente erforderlichen ANSYS-Elemente circu124 werden vereinbart und die zugehörigen Parameter in Real-Const. abgelegt.

```

et,4,circu124,2 !Kondensator
r,4,kapazi,u0
et,5,circu124,0 !Widerstand
r,5,rvor

```

Für die Kopplung von Schaltung und finiter Geometrie gibt es ein eigenes Element. Der Symmetriefaktor, hier 0,5 wird in einer Real-Const. übergeben.

```
et,6,circu124,6      !Koppelement
r,6,0.5              !Symmetriefaktor
```

Zur Darstellung der Schaltung auf dem Grafik-Fenster von ANSYS wird der Parameter **abm** benutzt. Für die Knoten der diskreten Schaltung werden ANSYS-Knoten angeordnet. (Geometrisch könnte man diese Knoten auch alle übereinander legen. Dann entfällt aber die geometrische Anordnungskontrolle.)

```
abm=0.1 !Abmessung für Schaltungsdarstellung
n,,-2*abm,0
n,,-2*abm,abm
n,,-abm,abm
n,,-abm,0
```

Nun werden die Elemente zwischen den Knoten generiert. Für das Koppelement ist als dritter Knoten einer des Leiters erforderlich.

```
type,4 $real,4
e,node(-2*abm,0,0),node(-2*abm,abm,0)
type,5 $real,5
e,node(-2*abm,abm,0),node(-abm,abm,0)
type,6 $real,6
e,node(-abm,abm,0),node(-abm,0,0),node(wiradin,0,0)
```





Abb. 4.3-2 Vernetzte Geometrie und speisende Schaltung

Die Elemente des Leiters (der Windung) müssen hinsichtlich der Freiheitsgrade **curr** und **emf** gekoppelt werden.

```
!---Kopplung-----
esel,s,mat,,2
nsle
cp,1,curr,all
cp,2,emf,all
```

Dann werden die unteren Knoten der diskreten Schaltung auf das Bezugspotential gelegt.

```
!-----Bezugspotential-----
nset,s,loc,x,-2*abm,-abm
nset,r,loc,y,0
d,all,volt,0
allset
```

Dann kann die transiente Lösung angefordert werden. Wenn die Induktivität der Windung nicht auf andere Weise ermittelt wurde und nichts über die zu erwartende Frequenz des Stromes ausgesagt werden kann, muss probiert werden. Hier wird der Zeitraum von 4 μs in 20 Schritten berechnet.

```
/solu
antype,trans
time,1e-12      !Startwert
solve
outres,all,all
time,4e-6
deltim,0.2e-6
solve
```

In erster Auswertung wird der Strom durch den Vorwiderstand ermittelt. Auf analoge Weise kann der Spannungsabfall und die Verlustleistung angegeben werden.

```
/post26
esel,s,type,,5
*get,er,elem,,num,min
esel,all
esol,2,er,,smisc,2,Strom
esol,3,er,,smisc,1,Spannung
esol,4,er,,nmisc,1,Leistung
/grid,1
/gthk,curve,2
/axlab,x,Zeit
/axlab,y,Strom
plva,2
```

Die Potentiale der beiden Schaltungsknoten sind folgendermaßen angebbbar.

```
nsol,2,node(-2*abm,abm,0),volt,,UC
nsol,3,node(-abm,abm,0),volt,,UC-UR
/axlab,y,Spannung
plva,2,3
```




Abb. 4.3-3 Strom durch den Widerstand




Abb. 4.3-4 Potentiale der Schaltungsknoten

Die weitere Auswertung der Stromdichteverteilung, der Lorentzkraft, der magnetischen Flussdichte und der magnetischen Feldstärke kann analog zu den anderen Beispielen erfolgen.